(HP-67/97) Hyperbolics and inverses in 56 steps
06-11-2020, 02:22 PM
Post: #1
 Dave Britten Senior Member Posts: 1,579 Joined: Dec 2013
(HP-67/97) Hyperbolics and inverses in 56 steps
I'm waiting for parts to fix the card reader in my 97, so I'm making a few short programs that can be keyed in quickly. This is adapted from the hyperbolic and inverse hyperbolic function programs in the HP-25 Applications Programs book, taking advantage of subroutines, flags, etc. to make the programs even shorter than the 25 versions.

The full program is 56 steps, and calculates hyperbolics and their inverses. If you don't need the inverses, the program is only 27 steps. For inverses only, the program is 31 steps (the combined program shares a few steps, hence the two individual programs are 58 steps in total).

Usage

With the function argument in x, use the following key sequences to calculate one of twelve functions:

sinh: A
cosh: B
tanh: C
sech: B 1/x
csch: A 1/x
coth: C 1/x

arcsinh: a
arccosh: b
arctanh: c
arcsech: 1/x b
arccsch: 1/x a
arccoth: 1/x c

Hyperbolics and Inverses

Code:
LBL A GSB 1 - GTO 2 LBL B GSB 1 + LBL 2 2 / RTN LBL C GSB 1 - ENTER ENTER LASTx 2 * + / RTN LBL 1 e^x ENTER 1/x RTN ==Stop here if you don't need inverses== LBL a CLF 0 GTO 1 LBL b STF 0 LBL 1 ENTER ENTER * 1 F? 0 CHS + SQRT + LN RTN LBL c ENTER ENTER 1 + x><y CHS 1 + / LN GTO 2

Inverses Only

Code:
LBL a CLF 0 GTO 1 LBL b STF 0 LBL 1 ENTER ENTER * 1 F? 0 CHS + SQRT + LN RTN LBL c ENTER ENTER 1 + x><y CHS 1 + / LN 2 / RTN
 « Next Oldest | Next Newest »

User(s) browsing this thread: 1 Guest(s)